
Integrity Check for the Epidemiology of Diabetes Interventions and Complications Study (EDIC) 
Four Year Retinopathy and Nephropathy Files 

 
As a partial check of the integrity of the EDIC Four Year Retinopathy and Nephropathy datasets archived 
in the NIDDK data repository, a set of tabulations was performed to verify that published results can be 
reproduced using the archived datasets. Analyses were performed to duplicate published results for the 
data reported by Lachin et al [1] in The New England Journal of Medicine in February 2000. The results 
of this integrity check are described below. The full text of The New England Journal of Medicine article 
can be found in Attachment 1, and the SAS code for our tabulations is included in Attachment 2. 
 
Background. The Diabetes Control and Complications Trial (DCCT), 1983-1993, was a multi-center trial 
designed to determine whether intensive therapy to maintain blood glucose and glycosolated hemoglobin 
concentrations as close to the normal range as possible would prevent or delay long-term complications in 
patients with Type 1 diabetes.  This trial showed a markedly reduced risk of microvascular complications 
as compared with conventional therapy.  Most participants were then enrolled in the EDIC study, a long-
term observational study.  One of the objectives of the EDIC study was to compare the long-term effects 
of the intensive or conventional therapy provided during the DCCT on the development of more advanced 
retinal and renal complications of diabetes.  The published data describes differences between the two 
original treatment groups in the incidence of these complications four years after the close of the DCCT 
[1]. 
 
In summary, in 1994, 96% of the DCCT participants were enrolled in EDIC for regular observational 
follow-up of metabolic and complications status.  Diabetes care was obtained from the EDIC participants’ 
own physicians [2]. 
 
Preliminary Tabulations. Initial tabulations of the archived datasets showed 16 participants without a 
treatment assignment.  The Data Coordinating Center (DCC) was notified of the discrepancy and reported 
that the treatment assignment variable on the retinopathy dataset is not the appropriate one to use.  
Instead, the treatment assignment variable found on the DCCT baseline dataset should be used for this 
analysis.  Use of this baseline variable yields accurate tabulations. 
 
Baseline Data. Table 1 of the 2000 New England Journal of Medicine article reports on baseline 
characteristics. Variables summarized in this baseline table (Table 1. Characteristics Of the 1208 Patients 
Enrolled in the EDIC Study Who Were Evaluated after Four Years of Follow-up) are taken from both 
analysis datasets (GEE_RET and RENEUR04) created for this study and the DCCT baseline dataset 
(BASELINE, for treatment assignment). Table A lists the variables used in our replication of the Table 1 
variables. 

Table A: Variables Used to Replicate Table 1 
 

Table 1 Variable Variables Used in Replication 

Sample size GEE_RET dataset: y4retpop (where y4retpop=1); 
BASELINE: group 

At DCCT entry  
   Women (%) GEE_RET dataset: sex 
   Age (yr) GEE_RET dataset: age 
   Duration of diabetes (yr) GEE_RET dataset: duration 
   Glycosolated hemoglobin (%) GEE_RET dataset: hbael 
At EDIC entry  
   Age (yr) GEE_RET dataset: closeage 
   Duration of diabetes (yr) GEE_RET dataset: closedur 
   DCCT follow-up (yr) GEE_RET dataset: dcctyr 

   Glycosolated hemoglobin (%) GEE_RET dataset: edicyear (where edicyear=0), 
dcct_hba† 

Level of retinopathy (%)  
  None GEE_RET dataset: dcct10 (where dcct10=1) 
  Microaneurysms only GEE_RET dataset: dcct20 (where dcct20=1) 
   Mild nonproliferative retinopathy GEE_RET dataset: dcct30 (where dcct30=1) 
   Moderate or severe nonproliferative        
    retinopathy 

GEE_RET dataset: dcct40, dcct50 (where dcct40=1 or 
dcct50=1) 

Photocoagulation during DCCT (%)  
   Scatter, for retinopathy GEE_RET dataset: dcctscat 
   Focal, for macular edema GEE_RET dataset: dcct_foc 
Nephropathy at EDIC entry (%)‡  
   Albumin excretion > 28 µg/min RENEUR04 dataset: dc_aer40 
   Albumin excretion > 208 µg/min RENEUR04 dataset: dcaer300 
   Creatinine clearance < 70 
ml/min/1.73m2 RENEUR04 dataset: dc_clr70 

Treatment at EDIC year 4 (%)  
   Continuous subcutaneous insulin            
    infusion (pump) or multiple daily           
    injections 

GEE_RET dataset: insreg 

   Self-monitoring of blood glucose ≥ 4      
    times per day GEE_RET dataset: sbgm_4 

† Variable is labeled as ‘Mean HbA1c during DCCT’.  DCC confirmed this is the variable to summarize   
    for glycosolated hemoglobin at EDIC entry. 
‡ Label is incorrectly written as ‘Nephropathy at EDIC year 3 or 4 (%)’ in published paper (as confirmed 
     by DCC). 
 
In Table B, we compare the results for characteristics calculated from the archived dataset to the results 
published in the 2000 New England Journal of Medicine article. As Table B shows, the results obtained 
from the archived data are the same as those in the published tabulations, with a few minor exceptions. 

Table B: Comparison of Baseline Table Values Computed in Integrity Check to Reference Article 
Values 

 
Treatment: Conventional Table 1 Variable 

Lachin et al 
(2000) 

Integrity Check Difference 

Sample size 603 603 0 
At DCCT entry    
   Women (%) 47 47 0 
   Age (yr) 27 ± 7 27 ± 7 0 
   Duration of diabetes (yr) 5.6 ± 4.1 5.6 ± 4.1 0 
   Glycosolated hemoglobin (%) 9.0 ± 1.6 9.0 ± 1.6 0 
At EDIC entry    
   Age (yr) 33 ± 7 33 ± 7 0 
   Duration of diabetes (yr) 11.7 ± 4.8 11.7 ± 4.8 0 
   DCCT follow-up (yr) 6.1 ± 1.7 6.1 ± 1.7 0 
   Glycosolated hemoglobin (%) 9.0 ± 1.2 9.0 ± 1.2 0 
Level of retinopathy (%)    
  None 18 18 0 
  Microaneurysms only 30 31 1 
   Mild nonproliferative retinopathy 30 30 0 
   Moderate or severe nonproliferative        
    retinopathy 22 22 0 

Photocoagulation during DCCT (%)    
   Scatter, for retinopathy 4 4 0 
   Focal, for macular edema 5 5 0 
Nephropathy at EDIC entry (%)†    
   Albumin excretion > 28 µg/min 13 13 0 
   Albumin excretion > 208 µg/min 3 3 0 
   Creatinine clearance < 70 
ml/min/1.73m2 1 1 0 

Treatment at EDIC year 4 (%)    
   Continuous subcutaneous insulin            
    infusion (pump) or multiple daily           
    injections 

75 74 1 

   Self-monitoring of blood glucose ≥ 4      
    times per day 40 40 0 

† Label is incorrectly written as ‘Nephropathy at EDIC year 3 or 4 (%)’ in published paper (as confirmed 
     by DCC). 

Table B: Comparison of Baseline Table Values Computed in Integrity Check to Reference Article 
Values (cont.) 

 
Treatment: Intensive Table 1 Variable 

Lachin et al 
(2000) 

Integrity Check Difference 

Sample size 605 605 0 
At DCCT entry    
   Women (%) 48 48 0 
   Age (yr) 27 ± 7 27 ± 7 0 
   Duration of diabetes (yr) 5.9 ± 4.2 5.9 ± 4.2 0 
   Glycosolated hemoglobin (%) 9.0 ± 1.6 9.0 ± 1.6 0 
At EDIC entry    
   Age (yr) 34 ± 7 34 ± 7 0 
   Duration of diabetes (yr) 12.1 ± 4.9 12.1 ± 4.9 0 
   DCCT follow-up (yr) 6.2 ± 1.7 6.2 ± 1.7 0 
   Glycosolated hemoglobin (%) 7.3 ± 0.9 7.3 ± 0.9 0 
Level of retinopathy (%)    
  None 29 29 0 
  Microaneurysms only 38 39 1 
   Mild nonproliferative retinopathy 22 22 0 
   Moderate or severe nonproliferative        
    retinopathy 11 11 0 

Photocoagulation during DCCT (%)    
   Scatter, for retinopathy 2 2 0 
   Focal, for macular edema 2 2 0 
Nephropathy at EDIC entry (%)†    
   Albumin excretion > 28 µg/min 7 7 0 
   Albumin excretion > 208 µg/min 2 2 0 
   Creatinine clearance < 70 
ml/min/1.73m2 2 2 0 

Treatment at EDIC year 4 (%)    
   Continuous subcutaneous insulin            
    infusion (pump) or multiple daily           
    injections 

95 95 0 

   Self-monitoring of blood glucose ≥ 4      
    times per day 45 45 0 

† Label is incorrectly written as ‘Nephropathy at EDIC year 3 or 4 (%)’ in published paper (as confirmed 
     by DCC). 

Table B: Comparison of Baseline Table Values Computed in Integrity Check to Reference Article 
Values (cont.) 

 
p-values Table 1 Variable 

Lachin et al 
(2000) 

Integrity Check Difference 

At DCCT entry    
   Women (%) 0.69 0.68 0.01 
   Age (yr) 0.13 0.12 0.01 
   Duration of diabetes (yr) 0.18 0.18 0 
   Glycosolated hemoglobin (%) 0.41 0.41 0 
At EDIC entry    
   Age (yr) 0.08 0.07 0.01 
   Duration of diabetes (yr) 0.11 0.10 0.01 
   DCCT follow-up (yr) 0.16 0.16 0 
   Glycosolated hemoglobin (%) <0.001 <0.001 0 
Level of retinopathy (%) <0.001 <0.001 0 
Photocoagulation during DCCT (%)    
   Scatter, for retinopathy 0.018 0.018 0 
   Focal, for macular edema 0.038 0.038 0 
Nephropathy at EDIC entry (%)†    
   Albumin excretion > 28 µg/min 0.002 0.002 0 
   Albumin excretion > 208 µg/min 0.14 0.13 0.01 
   Creatinine clearance < 70 
ml/min/1.73m2 0.63 0.63 0 

Treatment at EDIC year 4 (%)    
   Continuous subcutaneous insulin            
    infusion (pump) or multiple daily           
    injections 

<0.001 <0.001 0 

   Self-monitoring of blood glucose ≥ 4      
    times per day 0.064 0.053 

DCC confirmed 
typo.  0.053 is 
the correct p-
value. 

† Label is incorrectly written as ‘Nephropathy at EDIC year 3 or 4 (%)’ in published paper (as confirmed 
     by DCC). 

 
 

 



Norma Pugh                                           
June 15, 2007 

6 

Follow-up Data. Table 2 of the 2000 New England Journal of Medicine article reports on progression 
of retinopathy. Variables summarized in this follow-up table (Table 2. Progression Of Retinopathy 
Between The End Of The DCCT And After Four Years Of The EDIC Study, According To The DCCT 
Treatment Group) are taken from one analysis dataset (GEE_RET) and the DCCT baseline dataset 
(BASELINE, for treatment assignment). Table C lists the variables used in our replication of the Table 2 
variables. 
 

Table C: Variables Used to Replicate Table 2 
 

Table 2 Variable Variables Used in Replication 

Level of Retinopathy at End of DCCT 

GEE_RET dataset: y4retpop (where y4retpop=1), 
dcctscat (delete patients with dcctscat=1), 
stp3f_y4, edicscat (if stp3f_y4=1 or edicscat=1 
then “progression” occurred); BASELINE: group 

   All levels 
GEE_RET dataset: dcct10, dcct20, dcct30, dcct40, 
dcct50 (where dcct10=1 or dcct20=1 or dcct30=1 
or dcct40=1 or dcct50=1) 

   No retinopathy GEE_RET dataset: dcct10 (where dcct10=1) 
   Microaneurysms only GEE_RET dataset: dcct20 (where dcct20=1) 
   Mild nonproliferative retinopathy GEE_RET dataset: dcct30 (where dcct30=1) 
   Moderate nonproliferative retinopathy or 
worse 

GEE_RET dataset: dcct40, dcct50 (where 
dcct40=1 or dcct50=1) 

   Adjusted 

GEE_RET dataset: dcctetd (Logistic regression 
analysis was performed with adjustment for the 
severity of retinopathy at the end of the DCCT 
according to the Early Treatment Diabetic 
Retinopathy Study categories.) 

 
In Table D, we compare the follow-up results calculated from the archived dataset to the results published 
in the 2000 New England Journal of Medicine article. As Table D shows, the results obtained from the 
archived data are the same as those in the published tabulations, with a few minor exceptions. 

Table D: Comparison of Progression of Retinopathy Table Values Computed in Integrity Check to 
Reference Article Values 

 
Number of Patients Table 2 Variable 

Lachin et al (2000) Integrity Check Difference
All levels    
   Conventional therapy 581 581 0 
   Intensive therapy 596 596 0 
No retinopathy    
   Conventional therapy 109 109 0 
   Intensive therapy 173 173 0 
Microaneurysms only    
   Conventional therapy 184 184 0 
   Intensive therapy 233 233 0 
Mild nonproliferative retinopathy    
   Conventional therapy 178 178 0 
   Intensive therapy 132 132 0 
Moderate nonproliferative retinopathy or 
worse    

   Conventional therapy 110 110 0 
   Intensive therapy 58 58 0 

 
 

Progression of Retinopathy Table 2 Variable 
Lachin et al (2000) Integrity Check Difference

All levels    
   Conventional therapy 21 21 0 
   Intensive therapy 6 6 0 
No retinopathy    
   Conventional therapy 16 16 0 
   Intensive therapy 6 6 0 
Microaneurysms only    
   Conventional therapy 14 14 0 
   Intensive therapy 4 4 0 
Mild nonproliferative retinopathy    
   Conventional therapy 19 19 0 
   Intensive therapy 4 4 0 
Moderate nonproliferative retinopathy or 
worse    

   Conventional therapy 42 42 0 
   Intensive therapy 22 22 0 

Table D: Comparison of Progression of Retinopathy Table Values Computed in Integrity Check to 
Reference Article Values (cont.) 

 
Odds Reduction (95% CI) Table 2 Variable 

Lachin et al (2000) Integrity Check Difference
All levels 75 (64-83) 75 (63-83) 0 (1,0) 
No retinopathy 66 (26-84) 66 (24-85) 0 (2,1) 
Microaneurysms only 76 (49-88) 76 (47-89) 0 (2,1) 
Mild nonproliferative retinopathy 83 (60-93) 83 (56-94) 0 (4,1) 
Moderate nonproliferative retinopathy or 
worse 60 (18-80) 60 (17-81) 0 (1,1) 

Adjusted 72 (59-81) 72 (62-84) 0 (3,3) 
 
 

p-values Table 2 Variable 
Lachin et al (2000) Integrity Check Difference

All levels <0.001 <0.001 0 
No retinopathy 0.006 0.006 0 
Microaneurysms only <0.001 <0.001 0 
Mild nonproliferative retinopathy <0.001 <0.001 0 
Moderate nonproliferative retinopathy or 
worse 0.012 0.013 0.001 

Adjusted <0.001 <0.001 0 
 
Other Results. Remaining figures and follow-up tables from the 2000 New England Journal of 
Medicine article were spot checked and found to be consistent with replicated results.  The few 
discrepancies documented in this report are likely due to data corrections and updates made between the 
paper data freeze and the final data freeze.  The repository has high confidence in the integrity of the 
Retinopathy and Nephropathy analysis datasets. 

Notes 
 
1. Both analysis datasets (Retinopathy: GEE_RET and Nephropathy: RENEUR04) were examined in this 

replication analysis.  In addition to the analysis datasets, the DCCT baseline dataset, which is housed at 
the repository, should be used for treatment assignment. 

 ATTACHMENT 1 
 
 
 

John M. Lachin, Sc.D., Saul Genuth, M.D., Patricia Cleary, M.S., Matthew D. Davis, M.D. 
and David M. Nathan, M.D., Retinopathy And Nephropathy In Patients With Type I 

Diabetes Four Years After A Trial Of Intensive Therapy, The New England Journal of 
Medicine, February 2000; Volume 342, Number 6: 381-389. 

 
The full text of the article referenced will be provided to approved requestors along with 

the data archive. 
 



ATTACHMENT 2 
 

SAS Code for Baseline and Follow-Up Tabulations from EDIC Four Year Retinopathy and 
Nephropathy Datasets in the NIDDK Repository 

 

 
 
 



/***************************************************************************************/ 
/* 
/* Program: R:\05_Users\Norma\EDIC\NEJM-4yrRetinopathy\getdata.sas 
/* Author:  Norma Pugh 
/* Date:    27 April 07 
/* Purpose: Create SAS datasets from the SAS transport files provided to the repository. 
/***************************************************************************************/ 
/* NEW LOCATION FOR EDIC SAS FILES */ 
libname newlib 'R:\05_Users\Norma\EDIC\NEJM-4yrRetinopathy'; 
 
/* ORIGINAL LOCATION OF EDIC TRANSPORT FILES */ 
filename file1 'R:\03_Data_And_Tools\Database\Databases\DCCT EDIC\DCCT-EDIC_FINAL\NEW-
Studies\Four Year Retinopathy\edicREN4.xpt'; 
filename file2 'R:\03_Data_And_Tools\Database\Databases\DCCT EDIC\DCCT-EDIC_FINAL\NEW-
Studies\Four Year Retinopathy\edicRET4.xpt'; 
 
/* CREATE THE DATASETS */ 
proc cimport library=newlib infile=file1; run; 
proc cimport library=newlib infile=file2; run; 



/***************************************************************************************/ 
/* 
/* Program: R:\05_Users\Norma\EDIC\NEJM-4yrRetinopathy\table1.sas 
/* Author:  Norma Pugh 
/* Date:    15 June 07 
/* Purpose: Replicate Table 1 results from NEJM article: Retinopathy and Nephropathy in  
/*           Patients With Type 1 Diabetes Four Years After A Trial Of Intensive Therapy 
/*           (2000). 
/***************************************************************************************/ 
/************************/ 
/* Libnames and formats */ 
/************************/ 
libname trt  'R:\03_Data_And_Tools\Database\Databases\DCCT_EDIC\DCCT-
EDIC_FINAL\DCCT\Analyses\Baseline\SAS_DATA'; 
libname data 'R:\05_Users\Norma\EDIC\NEJM-4yrRetinopathy'; 
 
%include 'R:\03_Data_And_Tools\Database\Databases\DCCT_EDIC\DCCT-EDIC_FINAL\NEW-
Studies\Four Year Retinopathy\fmt_ret4.sas'; 
 
proc format; value yesno 0='0=no' 1='1=yes'; run; 
 
 
/*****************************/ 
/* Get treatment assignments */ 
/*****************************/ 
data trt; 
 set trt.baseline; 
 keep mask_pat group; 
run; 
 
proc sort data=trt; by mask_pat; run; 
 
 
/**************************************/ 
/* Get Table 1 population & variables */ 
/**************************************/ 
data table1; 
 set data.gee_ret(where=(y4retpop=1)); 
  
 /* Level of retinopathy at EDIC entry */ 
 length retinop $9 binary_retinop 3; 
 
 if dcct10=1 then retinop='1_none'; 
  else if dcct20=1 then retinop='2_micro'; 
  else if dcct30=1 then retinop='3_mild'; 
  else if dcct40=1 or dcct50=1 then retinop='4_mod_sev'; 
 
 if retinop='1_none' then binary_retinop=0; 
  else if retinop in('2_micro','3_mild','4_mod_sev') then binary_retinop=1; 
   
 /* Labels */ 
 label retinop        = 'Level of retinopathy at EDIC entry' 
       binary_retinop = 'Retinopathy @ EDIC entry (none vs oth)'; 
        
 drop group;        
run; 
 



proc sort data=table1; by mask_pat; run; 
 
 
/**********************************************/ 
/* Merge table data with treatment assignment */ 
/**********************************************/ 
data table1; merge table1(in=x1) trt(in=x2); by mask_pat; if x1 & x2; run; 
 
 
/* Characteristics at DCCT entry */ 
title'Table 1: Characteristics at DCCT Entry'; 
title2'Categorical Counts & p-values'; run; 
proc freq data=table1; tables group*sex / chisq; run; 
 
title2'Quantitative Means & Standard Deviations'; run; 
proc sort data=table1; by group; run; 
proc means data=table1 n mean std; 
 by group; 
 var age duration hbael; 
run; 
 
title2'Quantitative p-values'; run; 
proc npar1way data=table1 wilcoxon; 
 class group; 
 var age duration hbael;  
run; 
 
 
/* Characteristics at EDIC entry */ 
title'Table 1: Characteristics at EDIC Entry'; 
title2'Quantitative Means & Standard Deviations'; run; 
proc sort data=table1; by group; run; 
proc means data=table1 n mean std; 
 by group; 
 var closeage closedur dcctyr; 
run; 
 
title2'Quantitative p-values'; run; 
proc npar1way data=table1 wilcoxon; 
 class group; 
 var closeage closedur dcctyr;  
run; 
 
 
 /* Glycosylated hemoglobin (EDIC yr 0 results for the N=1208 study population) */ 
 data subj; set table1; keep mask_pat; run; 
 data hemo; set data.gee_ret; keep mask_pat group edicyear dcct_hba; run; 
 
 proc sort data=subj; by mask_pat; run; 
 proc sort data=hemo; by mask_pat; run; 
 
 data hemo; merge subj(in=in_popn) hemo; by mask_pat; if in_popn & edicyear=0; run; 
 
 title2'Quantitative Means & Standard Deviations'; run; 
 proc sort data=hemo; by group; run;  
 proc means data=hemo n mean std; 
  by group; 



  var dcct_hba; 
 run; 
 
 title2'Quantitative p-values'; run; 
 proc npar1way data=hemo wilcoxon; 
  class group; 
  var dcct_hba;  
 run; 
 
 
title2'Categorical Counts & p-values'; run; 
proc freq data=table1; tables group*retinop; run; 
proc freq data=table1; tables group*(binary_retinop dcctscat dcct_foc) / chisq; run; 
 
 
/* Nephropathy at EDIC entry */ 
title'Table 1: Nephropathy at EDIC Entry'; 
title2'Categorical Counts & p-values'; run; 
  
 /* Nephropathy data */ 
 proc sort data=data.reneur04 out=nephrop; by mask_pat; run; 
 
 /* Merge with treatment assisgnment data */ 
 data nephrop; merge trt nephrop(in=in_neph); by mask_pat; if in_neph; run; 
 
 /* Results */ 
 proc freq data=nephrop; tables group*(dc_aer40 dcaer300 dc_clr70) / chisq; run; 
 
 
/* Treatment at EDIC year 4 */ 
title1'Table 1: Treatment at EDIC year 4'; 
title2'Categorical Counts & p-values'; run; 
proc freq data=table1; tables group*(insreg sbgm_4) / chisq; run; 



/***************************************************************************************/ 
/* 
/* Program: R:\05_Users\Norma\EDIC\NEJM-4yrRetinopathy\table2.sas 
/* Author:  Norma Pugh 
/* Date:    15 June 07 
/* Purpose: Replicate Table 2 results from NEJM article: Retinopathy and Nephropathy in  
/*           Patients With Type 1 Diabetes Four Years After A Trial Of Intensive Therapy 
/*           (2000). 
/***************************************************************************************/ 
/************************/ 
/* Libnames and formats */ 
/************************/ 
libname trt  'R:\03_Data_And_Tools\Database\Databases\DCCT_EDIC\DCCT-
EDIC_FINAL\DCCT\Analyses\Baseline\SAS_DATA'; 
libname data 'R:\05_Users\Norma\EDIC\NEJM-4yrRetinopathy'; 
 
%include 'R:\03_Data_And_Tools\Database\Databases\DCCT_EDIC\DCCT-EDIC_FINAL\NEW-
Studies\Four Year Retinopathy\fmt_ret4.sas'; 
 
proc format; value yesno 0='0=no' 1='1=yes'; run; 
 
 
/*****************************/ 
/* Get treatment assignments */ 
/*****************************/ 
data trt; 
 set trt.baseline; 
 keep mask_pat group; 
run; 
 
proc sort data=trt; by mask_pat; run; 
 
 
/**************************************/ 
/* Get Table 2 population & variables */ 
/**************************************/ 
data table2; 
 set data.gee_ret(where=(y4retpop=1)); 
  
 /* Level of retinopathy at EDIC entry */ 
 length retinop $9; 
 
 if dcct10=1 then retinop='1_none'; 
  else if dcct20=1 then retinop='2_micro'; 
  else if dcct30=1 then retinop='3_mild'; 
  else if dcct40=1 or dcct50=1 then retinop='4_mod_sev'; 
 
 /* Exclude patients with scatter photocoagulation during the DCCT */ 
 if dcctscat=1 then delete; 
 
 /* Progressions */ 
 if stp3f_y4=1 or edicscat=1 then progress=1; else progress=0; 
 
 /* Total population */ 
 output; retinop='0_all'; output; 
  
 /* Labels */ 



 label retinop  = 'Level of retinopathy at EDIC entry' 
       progress = 'Progression of retinopathy'; 
 
 drop group;        
run; 
 
proc sort data=table2; by mask_pat; run; 
 
 
/**********************************************/ 
/* Merge table data with treatment assignment */ 
/**********************************************/ 
data table2; merge table2(in=x1) trt(in=x2); by mask_pat; if x1 & x2; run; 
 
 
/*****************************************************/ 
/* Number of patients and progression of retinopathy */ 
/*****************************************************/ 
title'Table 2: Progression of Retinopathy (End of DCCT to EDIC Year 4)'; 
proc freq data=table2 noprint; 
 tables retinop*group / out=denom(drop=percent rename=(count=denom)); 
run; 
 
proc print data=denom; title2'Total Number of Patients'; run; 
 
proc sort data=table2; by retinop; run; 
 
proc freq data=table2 noprint; 
 by retinop; 
 tables group*progress / out=counts(drop=percent); 
run; 
 
data counts; 
 merge counts(where=(progress=1)) denom; 
 by retinop group; 
 pct=(count/denom)*100; 
run; 
  
proc print data=counts; title2'Percent of Patients with Progression'; run; 
 
 
/***********************************/ 
/* Odds Reduction, 95% CI, p-value */ 
/***********************************/ 
%MACRO OR(category,out); 
 proc freq data=table2(where=(retinop="&category")) noprint; 
  tables group*progress / cmh1; 
  output out=&out cmh1; 
 run; 
 
 data &out; set &out; 
  odds_red=(1-(_mhrrc2_/_mhrrc1_))*100; /* MH Adjusted Col1 RR, MH Adjusted Col2 RR */ 
  low_ci=(1-(1/l_lgor))*100;    /* Lower CL, Logit Adjusted OR */ 
  high_ci=(1-(1/u_lgor))*100;    /* Upper CL, Logit Adjusted OR */ 
  pval=round(p_cmhcor,.0001);    /* P-value for CMH Nonzero Correlation */ 
  keep odds_red low_ci high_ci pval; 
 run;  



 
 proc print data=&out; TITLE2"&category Stats"; run; 
%MEND OR; 
 
%OR(0_all,all_stats); 
%OR(1_none,none_stats); 
%OR(2_micro,micro_stats); 
%OR(3_mild,mild_stats); 
%OR(4_mod_sev,modsev_stats); 
 
 
/********************************************/ 
/* Adjusted Odds Reduction, 95% CI, p-value */ 
/********************************************/ 
/* NOTE: Use 1/x for OR and CI */ 
proc logistic data=table2(where=(retinop^='0_all')) descending; 
 class group retinop; 
 model progress = group retinop dcctetd; 
run; 

